<!--#include file="./header.inc"-->
<%

select case aq
	case "getgenre" 		: 	call getGenre
	case "getgenreclass" 	: 	call getGenreClass
	case "getgenretype" 	: 	call getGenreType
	case "getviewlist" 		: 	call getViewList
	case else 				: 	call index
end select

call db.close()

function index()
	app.json "没有指定参数", 1
end function

'取得大类
function getGenre()
	dim arr
	arr = app.getGenreAllInfo()
	if isArray(arr) then
		app.json arr, 0
	else
		app.json "没有数据", 1
	end if
end function

'取得大类下的栏目分类
function getGenreClass()
	dim arr, j, k, data()
	arr = app.getGenreClass(iid)
	if isArray(arr) then
	    k = 0
		for each i in arr
			j = app.getClassInfo(i)
			if isArray(j) then
				redim preserve data(k)
				data(k) = j
				k = k + 1
			end if
		next
		app.json data, 0
	end if
	app.json "没有数据", 1
end function

'取得指定大类类型的大类
function getGenreType()
	dim arr, j, data()
	if iid = 0 then
	    app.json "请指定大类类型", 1
	end if
    arr = app.getGenreAllInfo()
    if isArray(arr) then
        j = 0
        iid = trim(iid)
        for each i in arr
            if trim(i(1)) = iid then
                redim preserve data(j)
                data(j) = i
                j = j + 1
            end if
        next
        if j > 0 then
            app.json data, 0
        end if
    end if
    app.json "没有数据", 1
end function

'取得模板列表
function getViewList()
	dim arr, data()
	arr = db.GetData("select id,viewName from cj_view")
	if isArray(arr) then
		for i = 0 to ubound(arr, 2)
			redim preserve data(i)
			data(i) = array(arr(0, i), arr(1,i))
		next
		app.json data, 0
	else
		app.json "没有数据", 1
	end if
end function

%>